// Copyright (C) 1993-2002, by Peter I. Corke
// Copyright (C) 2007  Interdepartmental Research Center "E. Piaggio", University of Pisa
//
// This file is part of RTSS, the Robotics Toolbox for Scilab/Scicos.
//
// RTSS is free software; you can redistribute it and/or modify
// it under the terms of the GNU General Public License as published by
// the Free Software Foundation; either version 2 of the License, or
// (at your option) any later version.
//
// RTSS is distributed in the hope that it will be useful,
// but WITHOUT ANY WARRANTY; without even the implied warranty of
// MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE.  See the
// GNU General Public License for more details.
//
// You should have received a copy of the GNU General Public License
// along with RTSS; if not, write to the Free Software
// Foundation, Inc., 51 Franklin St, Fifth Floor, Boston, MA  02110-1301  USA



function [w] = rt_maniplty(robot, q, which)
// File name:       rt_maniplty.sci
//
// Function:        rt_maniplty
//
// Description:     compute the scalar manipulability index for the manipulator at
//                  the given pose
//
// Annotations:     this code is a Scilab port of corresponding function in the
//                  Robotics Toolbox for MATLAB(R) written by Peter I. Corke.
//
// References:      Robotics Toolbox for MATLAB(R), robot7.1/maniplty.m
//
// Author:          Matteo Morelli, I.R.C. "E. Piaggio", University of Pisa
//
// Date:            April 2007
//
// $LastChangedDate: 2007-10-05 23:28:34 +0200(ven, 05 ott 2007) $

    [%nargout, %nargin] = argn(0);

    w = [];
    n = robot.n;
    if %nargin == 2 then
        which = "yoshikawa";
    end
    if length(q) == robot.n then
        q = q(:).';
    end

    if or(which == ["yoshikawa", "yoshi", "y"]) then

        for Q = q.',
            w = [w; rt_yoshi(robot, Q.')];
        end

    elseif or(which == ["asada", "asa", "a"]) then

        for Q = q.',
            w = [w; rt_asada(robot, Q.')];
        end

    end

endfunction
